Subtractive synthesis examples

1) Basic variable-frequency filter, with an option of three different inputs:

The first example is a very simple one. A reson filter with centre frequency varying (exponentially) from **p5** to **p6** Hertz. The filter has a constant **Q** (**p7**), which is used to calculate its bandwidth. The pfield 8 (**p8**) is used to select the source signal for the filter, 0 for external input, 1 for band-limited pulse, 2 for white noise input. The following lines

```
if isrc == 1 goto blp
if isrc == 2 goto noise
```

are used to select the source. For instance, if **1** is selected, the instrument jumps to the **blp** label:

```
blp:
ain  buzz  p4, 100, 40, 1 ; source = 1, 100Hz band-limited pulse
goto filter
```

and after the **ain** variable is set, jumps again to the **filter** label, which is the rest of the instrument. If 2 is selected a similar process will happen. If neither 1 or 2 are selected, the instrument carries on and finds the external audio input lines

```
ain in                  ; source = 0, sound input
goto filter
```

The filter itself has its frequency/BW controlled by **kcf**, which is an exponential line between 1 and the ratio between the two frequencies **ifr2:ifr1**. This signal when multiplied by **ifr1** will create a curve between **ifr1** and **ifr2**.

Here's the complete instrument. Can you supply a score for it ?

```
; variable frequency filter
; Victor Lazzarini, 1998

sr= 44100
kr = 4410
ksmps = 10
nchnls= 1

instr 1

ifr1   =  p5     ; p5, p6 = filter freqs
ifr2   =  p6
iq     =  p7     ; Q
isrc   =  p8     ; p8 = source
idur   =  p3     ; duration


if isrc == 1 goto blp
if isrc == 2 goto noise

ain in                  ; source = 0, sound input
goto filter

blp:
ain  buzz  p4, 100, 40, 1 ; source = 1, 100Hz band-limited pulse
goto filter

noise:
ain  rand  p4             ; source = 2, white noise
goto filter

filter:
kcf   expon    1, idur, ifr2/ifr1  ; freq glissando
aflt  reson    ain, kcf*ifr1, (kcf*ifr1)/iq ; filter
```

```
aout  balance  aflt, ain          ; balance output

        out aout


endin
```

The next two instruments are adapted from Dodge & Jerse, but the page numbers refer to the 1st edition of the book. Nevertheless, you will be able to find these examples in the chapter on Subtractive Systems. Please refer to that text for comments/explanations. Please write suitable scores for them.

2) Glissandoing noise bands

```
; Adapted from Dodge & Jerse, pp.171-2
; Victor Lazzarini, 1998

sr= 44100
kr = 4410
ksmps = 10
nchnls= 1

instr 1

iamp = p4        ; noise src amplitude
ifr1   =  p5     ; filter frequencies
ifr2   =  p6
iminbw =  p7     ; minimum % BW
irange =  p8     ; BW range

;freq gliss & BW enlargement
kcfl  line    1, p3, ifr2/ifr1       ; generates a line from 1 to ifr2:ifr1 ratio
kcf   pow     kcfl, 2, ifr2/ifr1     ; power of 2 + normalization transforms
                                     ; the glissando into an exponential one
kbw   line    0, p3, irange

; noise source, filtering and balance
aran  rand     iamp
aflt  reson    aran, kcf*ifr1, (kbw+iminbw)*(kcf*ifr1),2 ; filter and scale
                                                         ; output to overall RMS input
      out aflt
endin
```

3) Timbre exchange instrument

```
; Timbre exchange instrument, adapted from Dodge & Jerse, p117-8
; Victor Lazzarini, 1998

sr = 44100
kr = 4410
ksmps = 10
nchnls= 1

instr 1

iamp  = p4           ; source amplitude
ifund  = cpspch(p5) ; fundamental
iq1    = p6         ; filter bank Q's
iq2    = p7

; band-limited pulse producing harmonics up to 6000 Hz
abuz  buzz  p4, ifund, 6000/(ifund), 1

; filterbank 1
af11   reson  abuz,  200,  800/iq1
af12   reson  af11,  800, 1300/iq1
af13   reson  af12, 1300, 1700/iq1
af14   reson  af13, 1800, 2000/iq1
af15   reson  af14, 2500, 2500/iq1
```

```
; filterbank 2
af21   reson  abuz, 1700, 1700/iq2
af22   reson  af21, 2000, 2000/iq2
af23   reson  af22, 2800, 2800/iq2
af24   reson  af23, 4000, 4000/iq2
af25   reson  af24, 5900, 5000/iq2

; signal balance
abal1   balance af15, abuz
abal2   balance af25, abuz

; timbral exchange
asig1    oscil  abal1, 1/p3, 2
asig2    oscil  abal2, 1/p3, 3

;hi-pass for emphasis, envelope
ahp     reson   asig1+asig2, sr/2, 10000
aout    linen   ahp, .05, p3, .09

         out     aout

endin

;score function tables 2 & 3

f2 0 1025 5 1 256 1 512 0 256 0
f3 0 1025 5 0 256 0 512 1 256 1
```